
News  
Signal version 4  

We are pleased to announce the release of Signal version 4. A list of the new features is 
included in the Signal section below. We will be writing to all registered users of Signal 
with upgrade details. 

We support running version 4 under Windows 98SE, ME, NT2000, XP and Vista. It will 
run on Windows 95, but we do not support this as we no longer maintain a Windows 95 
system. If you have a choice of operating system, we would recommend that you use 
Windows XP. 

The release of version 4 means that we will not be adding significant new features to 
version 3, but we will continue to fix bugs. 

CED at Society for Neuroscience 2007 
We will be exhibiting at the Society for Neuroscience meeting in San Diego as well as 
running our Spike2 and Signal training courses on Thursday 1st and Friday 2nd of 
November. For those of you who will not see us at the training days, we hope to see you 
at booth 1620 during the course of the meeting. 

Meetings and events 
Society for Neuroscience 2007                                                                                                                   
Booth 1620 
San Diego, California  
November 3rd – 7th 2007 

American Heart Association - Scientific Sessions                                                                                                           
Booth 1835 
Orlando, Florida 
November 3rd – 6th 2007 

 
Latest versions of Spike2 and Signal 

Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.05  Signal version 3.10 

 Spike2 version 5.19  Signal version 2.16 

 Spike2 demo  Signal demo 

Spike2   

 

Q. I have recently started using the spike sorting features in Spike2, which works 
well with my nerve recordings, where the spikes seem to stay constant in 
amplitude. However some of my future recordings will be based on drug 
responses where the spike amplitude of individual units will probably change 
over time. How does Spike2 deal with this kind of data? 

A. There are a number of settings that control the formation and matching of spikes 
 to templates which can be accessed from the Template parameters menu item in 
 the WaveMark window. Spike2 can scale each captured spike up and down by 
 the value specified in the Maximum percent amplitude change for match field 
 while matching spikes to existing templates, or track the shape of the spike by 
 modifying the template for each spike added. This can be effective for slow or 
 gradual changes in spike amplitude. If the changes in spike amplitude are 
 sudden then the better option is to allow Spike2 to generate multiple templates 
 for the spike and then simply change the code of the multiple templates to 
 match the original template. You can change the code of a template simply by   
 double-clicking on the corresponding number (circled below) in the template area. 

 
WaveMark and Template parameters dialog 

You can also change template codes interactively if the spikes are displayed in 
Overdraw WM mode by holding down Alt+Ctrl and left-clicking the mouse to 
draw a line. Any spike that crosses the line in the display can then be re-
classified according to the code entered in the dialog. 

 

 
Editing marker codes in Overdrawn WaveMark mode 

The size of the template windows can be changed between small, large and huge by double-clicking in the template area

Q. Can you give me a description of some of the new features that will be available 
in Signal version 4? 

A. Version 4 is the latest release of Signal and includes many new and improved 
features. 

 These include: 

• A new measurements system allows active cursors to iterate through 
multiple features in a single frame 

• Gap-free sampling mode allows continuous sampling across multiple 
frames 

• Variable sweep points per frame controlled by multiple states 

• Interactive IIR digital filtering 

• SCAN method of idealised trace generation – based on an algorithm 
developed by Professor David Colquhoun at University College London -  
now includes incremental processing of trace formation 

• New dialog allows customization of the keyboard shortcuts used to 
control the idealised trace editor dialog 

• Maximum likelihood exponential fits for Open/Closed time histograms 

• New script editor with code folding, auto formatting, auto complete and 
call tips 

  Files created by previous versions of Signal are fully compatible with version 4. 
 As with previous versions, future revisions of version 4 will be freely available to 
 download from our website for all registered version 4 users. 

 

 Scripts: Spike2  

Q. We are recording EEG activity and would like to automatically monitor and detect 
responses to drug application based on the total power FFT between 5Hz and 
40Hz of the EEG waveform. Before drug application we would like to determine a 
baseline threshold for the total power content of the EEG based on the FFT 
result from the last 30 seconds of data, split up into sections of n seconds 
(determined by the user). The FFT should be performed on each section of data 
and the maximum value detected should be set as the threshold level. Once the 
drug has been applied we would like to monitor the FFT of the incoming data 
(again for each time section as previously specified) and output a pulse once the 
threshold activity has been exceeded. The monitoring of the FFT content should 
continue until the activity drops below the threshold, when another output pulse 
should be triggered.    

A. The attached script FFTMonitor.s2s along with the sequence file 
FFTMonitor.pls, is an example of monitoring incoming data and responding to 
changes. The sequence file should be loaded into your existing sampling 
configuration before running the script. Once sampling has started, the script can 
determine a threshold value for the total power between 5Hz and 40Hz based on 
the last 30 seconds of data, split into sections. The user can enter the time of the 
individual sections and the settings for the FFT in a dialog. The script can then 
monitor the incoming data based on this threshold level. Once the total power 
exceeds the threshold, a pulse is output from the digital output 0 port of the 1401. 
When the signal drops back below the threshold level a second pulse is output 
from digital output 1. The user can decide when to determine the threshold value 
and when to start monitoring the incoming waveform data using the script toolbar. 

 

 Scripts: Signal  

Q. Is there any way to take a measurement of the area over baseline where the 
baseline can be adjusted interactively? 

Back to top

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$FFTMonitor|Script to calculate power content of incoming cycles of waveform data and respond to a threshold crossing

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 

var data%;								'Handle of new data file (Spike2 sets it to 0)
var sTime;								'The last time we looked at the idle routine
var stim;
var pSpec%;
var thresh;
var consCyc%;
var flag%;

var cycle:=0.3;
var chan%:=1;
var fftSz%:=512;
var wnd%:=1;
var ind%;
var sizes%[9];
var dlgqq$[9];
var wind$[10];


sizes%[0] := 16; 
sizes%[1] := 32; 
sizes%[2] := 64; 
sizes%[3] := 128;
sizes%[4] := 256; 
sizes%[5] := 512; 
sizes%[6] := 1024;
sizes%[7] := 2048; 
sizes%[8] := 4096;

dlgqq$[0] := "16";
dlgqq$[1] := "32";
dlgqq$[2] := "64";
dlgqq$[3] := "128";
dlgqq$[4] := "256";
dlgqq$[5] := "512";
dlgqq$[6] := "1024";
dlgqq$[7] := "2048";
dlgqq$[8] := "4096";

wind$[0] := "None";
wind$[1] := "Hanning";
wind$[2] := "Hamming";
wind$[3] := "Kaiser 30 dB";
wind$[4] := "Kaiser 40 dB";
wind$[5] := "Kaiser 50 dB";
wind$[6] := "Kaiser 60 dB";
wind$[7] := "Kaiser 70 dB";
wind$[8] := "Kaiser 80 dB";
wind$[9] := "Kaiser 90 dB";


HideAll();                       'Hide all unwanted bars
ToolbarVisible(1);					'Make toolbar visible always
New%();									'Set up new sampling window
DoToolbar();							'Do the toolbar
RestoreAll();                    'Restore all bars
Halt;

func New%()								'New sampling window
View(LogHandle());					'Make log view the current view
EditSelectAll();						'Select all text in log view
EditClear();							'Delete it
Window(0,80,100,100);				'Display it at the bottom of the screen
WindowVisible(1);						'Make it visible
SampleDigMark(100);

if data%>0 then						'If there is already a data view open...
  View(data%);							'...close it
  FileClose();                   'No point setting data% to 0...
endif;

data%:=FileNew(0,1);					'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);						'Set event draw mode to lines
Window(0,0,100,80);					'Make data window in top bit of screen
XRange(0,10);
ChanHide(32);

FrontView(LogHandle());				'Bring the Log view to the front
FrontView(data%);						'Bring the data view to the front
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarSet(6,"Determine Threshold",GetThresh%);
ToolbarSet(8,"Monitor seizure",Monitor%);
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(6,0);
ToolbarEnable(8,0);

Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
sTime := 0;                      'set start of anaylsis time
SampleStart();							'Start sampling
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarEnable(6,1);
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

Func GetThresh%()
var psCycle%;
var cTime;
var start,fin;
var noSteps%;
var c%;
var IniDelay;
var pulseDelay;

if MaxTime() > 30 then          'Check that we have enough previous data

DlgCreate("Power Spectrum Settings");
DlgChan(1,"Select Channel",1);
DlgList(2,"FFT block size",dlgqq$[],11);
DlgList(3,"Window",wind$[],10);
DlgReal(4,"Section time (s)",0.01,10);      'Set FFT settings
DlgShow(chan%,ind%,wnd%,cycle);

fftSz%:= sizes%[ind%];

cTime:= MaxTime();
start:= (MaxTime())-30;     'get 30 seconds of data
fin:= start+cycle;

psCycle%:= SetPower(chan%,fftSz%,wnd%);
CursorNew();
Cursor(1, XToBin(5));  'Fetch the cursor
CursorNew();
Cursor(2, XToBin(40));  'Fetch the cursor

noSteps%:= 30/cycle;
var psSum[noSteps%+1];      'Calculate no. of sections

for c%:=1 to noSteps% do
    Process(start,fin,1);
    psSum[c%]:= ChanMeasure(1,4,BinToX(Cursor(1)),BinToX(Cursor(2)));   'Get total power for each section
    
    start+=cycle;
    fin+=cycle;
next; 

ArrSort(psSum[],1);     'Sort to find the maximum value
thresh:= psSum[1];

    if thresh = 0.0 then
        Message("Not enough data in a section to calculate threshold\nTry reducing the FFT block size!");       'If no. of points in section not enough
    else
        Message("Threshold level = %f",thresh);

        IniDelay:= (cycle*1000)-3;          'Calculate delay
        pulseDelay:= ((cycle)*1000)-47;     'and pulse delay
        SampleSeqVar(1,iniDelay);
        SampleSeqVar(2,pulseDelay);         'For cycle marker
        ToolbarEnable(8,1);             
    endif;
else
    Message("Not enough data recorded! 30 seconds of elapsed\ndata is required to calculate the activity threshold");
endif;
return 1;
end;

Func Monitor%();        'Monitor function
View(data%).Window(0,0,50,80);
pSpec%:=SetPower(chan%,fftSz%,wnd%);            'Generate power spectrum window
WindowVisible(1);
ProcessAuto(cycle,1,1,cycle);       'Set process based on cycle marker

Window(50,0,100,80);
XRange(0,50);
CursorNew();
Cursor(1, XToBin(5));  'Fetch the cursor
CursorNew();
Cursor(2, XToBin(40));  'Fetch the cursor
ToolbarSet(0,"",Idle%);			'Call Idle%() whenever there is free time
return 1;
end;

func Idle%()							'Idle routine runs when system has time
var eTime;
var tp;

if ViewKind(data%)=0 then        'This checks that the data view exists...
   View(data%);                  '...as this gives an error if it doesn't
   eTime:=Maxtime(); 

    if (eTime > sTime) then
        stim:=NextTime(32,sTime);    'Look for markers on memory channel
   
        if stim>sTime then				'If stimuli is found
            FrontView(pSpec%);
            tp:= ChanMeasure(1,4,BinToX(Cursor(1)),BinToX(Cursor(2)));  'Take measurement
            
            if flag%=0 then
                if tp > thresh  then        'Check for TP above threshold
                    'consCyc%+=1;
                    'if consCyc% > 15 then       
                        SampleKey("a");     'Output pulse
                        flag%:=1;           'Set flag for above threshold
                        consCyc%:=0;
                        'endif;
                else
                    consCyc%:=0;
                endif;
             else
                if tp < thresh then         'Check for TP below threshold
                    'consCyc%+=1;
                    'if consCyc% > 15 then
                        SampleKey("b");     'Output pulse
                        flag%:=0;           'Set flag fro below threshold
                        consCyc%:=0;
                    'endif;
                else
                    consCyc%:=0;
                endif;
            endif;
            
            PrintLog("%f\n",tp);        'Print values to the Log
        endif;
        sTime:=eTime;       'record current time
      endif;
   endif;
return 1;								'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];              'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

'$Area over baseline.sgs|Signal script to calculate area over a specified baseline

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 

var data%;
var chan% := 1;
var o1,o2,oh;


data% := FileOpen("",0,2);      'Prompt user to open file
ToolbarText("");
Window(0,0,100,100);        'Make full screen
FrontView(data%);               'Bring to the front
SetChan%();
CursorSet(2);               'Set 2 vertical sursors
HCursorDelete(-1);          'Clear any existing horizontal cursors
HCursorNew(chan%);      'Set to channel

ToolbarSet(1, "Quit");      'Button to Quit
ToolbarSet(2, "Set Channel",SetChan%);      'Button to set channel to measure
ToolbarSet(0, "Idle", Idle%);       'Idle routine
Toolbar("",1023);


Func Idle%()
View(data%);
if Cursor(1) <> o1 or Cursor(2) <> o2 or HCursor(1) <> oh then      'If any cursor position has changed
	GetArea(chan%,Cursor(1),Cursor(2),HCursor(1));      'Take measurement
endif;
return 1;
end;

Func SetChan%()
DlgCreate("Set Channel");               'Set channel to measure
DlgChan(1,"Channel to measure",1);
DlgShow(chan%);         
HCursorDelete(-1);          'Clear any existing horizontal cursors
HCursorNew(chan%);      'Set to channel
return 1;
end;

Proc GetArea(ch%,sTime,eTime,lev)
var pts%,item%;
var area$,areaBase;
pts% := ChanRange(ch%,sTime,eTime,item%);       'Get no of data points in range
var data[pts%];
ArrConst(data[],View(data%,ch%).[item%:pts%]);  'Copy data to array
ArrSub(data[],lev);         'Subtract the baseline level
areaBase := ArrSum(data[])*BinSize(ch%);    'Calculate the area
area$ := Str$(areaBase);
ToolbarText(area$);     'Display in toolbar
o1 := sTime;    'Remember positions
o2 := eTime;
oh := lev;
end;

A. The attached script, AreaOverBaseline.sgs, is an example of using an idle 
routine to update measurements based on changes in cursor positions. It allows 
the user to position a horizontal cursor at the baseline position and two vertical 
cursors around the area to measure. The area over baseline measurement is 
then displayed in the script toolbar and updated whenever the baseline or start 
and end position are changed. 

Did you know…? 
You can display X axis units for Spike2 data files in seconds, hours, minutes and 
seconds or as time of day relative to the operating system clock from the X Axis Range 
dialog. 

 
X Axis Range dialog 

Recent questions   
Q. Is there any way to specify where the temporary data file is located when 

sampling with Spike2? I have a separate physical drive that I’d like to use 
explicitly for capture of large data files, rather than using the C:\ drive.  

A. You can set the directory path for the temporary data file from the Edit 
Preferences menu in Spike2. The Sampling tab of the dialog lets you browse to a 
location or enter the location as a text string. For efficient data transfer this 
should always be a physical hard drive on the host PC, rather than a network 
drive.  

 
Edit preferences menu 

  

     

User group

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

